

Documentation création et déploiement automatisé d'une image



Pour créer une image complète de l'ordinateur qui va servir d'image, il faut utiliser DISM (Deployment Imaging and Servicing Management) qui va permettre la mise au point d'images de déploiement.

Pour cela, il faut effectuer la commande suivante dans l'environnement de récupération Windows pour créer l'image (Paramètres→Windows Update→Options avancées→Récupération→Démarrage avancé→Dépannage→Options avancées→Invite de commandes) :

```
dism /Capture-Image /ImageFile:[Disque externe]:\[chemin vers  
l'image].wim /CaptureDir:C:\ /Name:"[nom indicatif pour l'image]"  
/Compress:fast /CheckIntegrity
```

Exemple : `dism /Capture-Image /ImageFile:D:\Images\imagepc.wim
/CaptureDir:C:\ /Name:"IMAGE PC" /Compress:fast /CheckIntegrity`

Remarque : S'il y a une erreur de type "Espace insuffisant sur le disque", cela veut dire qu'il n'y avait pas assez d'espace pour les fichiers temporaires sur C:\.

Pour y remédier, créez un répertoire dans le disque externe pour ces fichiers temporaires et rajoutez la fonction permettant d'indiquer que les fichiers temporaires doivent être stockés sur le répertoire prévu à cet effet :


```
mkdir [Disque externe]:\Temp  
dism /Capture-Image /ImageFile:[Disque externe]:\[chemin vers  
l'image].wim /CaptureDir:C:\ /Name:"[nom indicatif pour l'image]"  
/Compress:fast /CheckIntegrity /ScratchDir:[Disque externe]:\Temp
```

Maintenant que l'image est créée, il faut la déployer sur plusieurs postes.

Pour cela, on peut utiliser une clé USB bootable avec WinPE.

Il faudra pour cette solution télécharger le [Windows ADK](#) et le [WinPE Add-on](#) depuis le site de Microsoft.

Pour Windows ADK, cocher uniquement la case “Outils de déploiement” comme suit :





Une fois les 2 outils installés, il faut aller dans l'environnement de déploiement et d'outils de création d'images en tant qu'administrateur :



Environnement de déploiement et d'outils de création d'images

Application

 Ouvrir

 Exécuter en tant qu'administrateur

Il faudra ensuite monter l'environnement WinPE :

```
cotype amd64 C:\WinPE_amd64
```

Puis copier les fichiers WinPE sur une clé USB préalablement formatée en fat32 :

```
xcopy C:\WinPE_amd64\media\* [Lettre de la clé USB]:\ /s /e
```

Exemple : *xcopy C:\WinPE_amd64\media* D:\ /s /e*

Enfin, il faut rendre la clé bootable :

```
bootsect /nt60 [Lettre de la clé USB]:
```

Exemple : *bootsect /nt60 D:*

Maintenant que la clé est prête, il faut ajouter 3 fichiers pour automatiser l'installation de l'image : startnet.cmd, ScriptAutoDeployIMG.bat et CommandesDiskpart.txt

Le script startnet.cmd est le script de démarrage par défaut de l'environnement WinPE.

Ce fichier va permettre dès lors que la clé sera bootée, d'exécuter un script de démarrage qui va appeler le script ScriptAutoDeployIMG.bat pour qu'il s'exécute juste après.

Cependant, ce script est inclus dans le boot.wim qui se situe dans la clé qu'on a créé précédemment dans le répertoire "\sources".

C'est pourquoi il faut pouvoir remplacer ce script par le nôtre.

Pour ce faire, aller dans une invite de commande en mode administrateur et créez un répertoire dans lequel on va monter le boot.wim afin de modifier le contenu de startnet.cmd :

```
mkdir C:\WinPE_Mount  
dism /Mount-Image /ImageFile:"[lettre de la clé USB]:\sources\boot.wim"  
/Index:1 /MountDir:C:\WinPE_Mount
```

Exemple : `dism /Mount-Image /ImageFile:"D:\sources\boot.wim"
/Index:1 /MountDir:C:\WinPE_Mount`

Il faudra ensuite aller dans "C:\WinPE_Mount\Windows\System32\" et modifier le script :

```
cd C:\WinPE_Mount\Windows\System32\  
notepad startnet.cmd
```

Le script à mettre :

```
@echo off
wpeinit
```

```
:: Trouver la cle USB contenant le script d installation
for %%D in (D E F G H I J K L M N O P Q R S T U V W Y Z) do (
    if exist %%D:\ScriptAutoDeployIMG.bat (
        set USB=%%D:
        echo Cle USB detectee !
        goto usb_found
    )
)

:: Si la cle n est pas trouvee, erreur
echo ERREUR : Impossible de trouver ScriptAutoDeployIMG.bat sur la
cle USB !
pause
exit

:usb_found
:: Lancer le script principal
call %USB%\ScriptAutoDeployIMG.bat
exit
```

Enfin, sortez du répertoire “*WinPE_Mount* ” ,démontez l'image que l'on vient de monter et supprimez le répertoire (Fermer tous les fichiers en rapport avec ce répertoire pour éviter une erreur !) :

```
cd C:\
dism /Unmount-Image /MountDir:C:\WinPE_Mount /Commit
rmdir WinPE_Mount
```

Maintenant que le script startnet.cmd est fait, passons au script ScriptAutoDeployIMG.bat.

Ce script va permettre d'automatiser l'installation de l'image et doit être à la racine de la clé USB (comme mentionné dans le script précédent).

Le script à mettre (exemple pour une image s'appelant imageop.wim) :

```
@echo off
title Installation Automatique
echo =====
echo ===  INSTALLATION AUTOMATIQUE  ===
echo =====
echo.

:: Verification que la cle USB est bien detectee
if not defined USB (
    echo ERREUR : La cle USB n a pas ete detectee.
    pause
    exit
)

:: etape 1/5 : Partitionnement du disque
echo [1/5] Partitionnement du disque...
diskpart /s %USB%\CommandesDiskpart.txt

:: etape 2/5 : Recherche du disque contenant l image WIM
echo [2/5] Recherche du disque contenant imageop.wim...
set DISKEXT=
for %%D in (C D E F G H I J K L M N O P Q R S T U V W Y Z) do (
    if exist %%D:\imageop.wim (
        set DISKEXT=%%D:
```

```

        echo Image trouvee sur %DISKEXT%\imageop.wim
        goto image_found
    )
)

:: Si l'image n'est pas trouvee, erreur
echo ERREUR : Impossible de trouver imageop.wim sur un disque
externe !
pause
exit

:image_found
:: etape 3/5 : Application de l'image Windows
echo [3/5] Application de l'image Windows...
dism /apply-image /imagefile:%DISKEXT%\imageop.wim /index:1
/applydir:C:\

:: etape 4/5 : Configuration du boot UEFI
echo [4/5] Configuration du boot UEFI...
bcdboot C:\Windows /s S: /f UEFI

:: etape 5/5 : Redemarrage automatique
echo [5/5] Redemarrage du systeme...
wpeutil reboot

```

Enfin, le fichier CommandesDiskpart.txt va permettre au script ScriptAutoDeployIMG.bat de pouvoir partitionner le disque afin d'avoir une partition système dans lequel les fichiers de boot seront créés et une partition principale contenant l'image appliquée.

Le texte à mettre :

```

select disk 0
clean
convert gpt

```



```
rem Création de la partition EFI (100 Mo)
create partition efi size=100
format fs=fat32 quick
assign letter=S
```

```
rem Création de la partition principale (Windows) sans taille pour
occuper tout l'espace
create partition primary
format fs=ntfs quick
assign letter=C
```

```
exit
```

Maintenant que tout cela est fait, il faut aller dans le BIOS de la machine à déployer (Basic Input/Output System) et désactiver le Secure Boot. Il faut ensuite brancher la clé bootable et le disque externe contenant l'image.

Entrez dans le "Boot Mode" en appuyant de façon répétée sur F12 (Lenovo) et sélectionnez la clé bootable créée précédemment.

Dès que l'ordinateur aura redémarré, enlevez la clé et le disque externe et le PC pourra booter sur l'image appliquée !